

mgr Joanna UDALSKA

```

100 REM DATA KALENDARZOWA <---> DZIEŃ JULIANSKI
110 INPUT "DATA KALENDARZOWA (0) LUB DZIEŃ JULIANSKI (1) ";K
120 IF K=0 THEN GOTO 330
130 PRINT "DATA Z KALENDARZA JULIANSKIEGO (KJ) LUB"
140 PRINT "GREGORIANSKIEGO (KG) --> DZIEŃ JULIANSKI"
150 INPUT "KJ(0) LUB KG(1) ";G
160 INPUT "ROK, MIESIĄC, DZIEŃ ";R,M,D
170 INPUT "CZAS UNIWERSALNY (GODZ, MIN, SEK) ";GO,MI,SE
180 LET FF=((60*GO+MI)*60+SE)/86400
190 LET F=FF-.5
200 LET J=-INT(7*(INT((M+9)/12)+R)/4)
210 IF G=0 THEN GOTO 260
220 LET S=S6N(M-9)
230 LET A=ABS(M-9)
240 LET J1=INT(R+S*INT(A/7))
250 LET J1=-INT((INT(J1/100)+1)*3/4)
260 LET J=J+INT(275*M/9)+D+6*J1
270 LET J=J+1721027!+2*G+367*R
280 IF F>=0 THEN GOTO 310
290 LET F=F+1
300 LET J=J-1
310 PRINT "DZIEŃ JULIANSKI: ";J;F
320 IF K=1 THEN GOTO 630
330 PRINT "DZIEŃ JULIANSKI --> DATA Z KALENDARZA"
340 PRINT "JULIANSKIEGO (KJ) LUB GREGORIANSKIEGO (KG)"
350 INPUT "KJ(0) LUB KG(1) ";G
360 INPUT "DZIEŃ JULIANSKI, ULAMEK ";J,F
370 LET F=F+.5
380 IF F<1 THEN GOTO 410
390 LET F=F-1
400 LET J=J+1
410 IF G=1 THEN GOTO 430
420 LET A=J: GOTO 450
430 LET A1=INT((J/36524.25)-51.12264)
440 LET A=J+1+A1-INT(A1/4)
450 LET B=A+1524
460 LET C=INT((B/365.25)-.3343)
470 LET D=INT(365.25*C)
480 LET E=INT((B-D)/30.61)
490 LET D=B-D-INT(30.61*E)+F
500 LET M=E-1
510 LET R=C-4716
520 IF E>13.5 THEN M=M-12
530 IF M<2.5 THEN R=R+1
540 LET D1=INT(D)
550 LET H=D-D1
560 LET GO=INT(24*H)
570 LET HH=24*H-GO
580 LET MI=INT(60*HH)
590 LET HHH=60*HH-MI
600 LET SE=INT(60*HHH)
610 PRINT "DATA (ROK, MIESIĄC, DZIEŃ) ";R,M,D1
620 PRINT "CZAS UNIWERSALNY (GODZ, MIN, SEK) ";GO,MI,SE
630 END

```

Podstawową jednostką większości kalendarzy stosowanych w dziejach ludzkości jest czas, w którym dokonuje się pełna zmiana pór roku — tzw. rok zwrotnikowy. Nie jest on jednak całkowitą wielokrotnością dni, wobec czego żaden kalendarz nie może być doskonały — nie może dawać pełnej zgodności z ruchem Słońca.

Kalendarz juliański wprowadzony w Rzymie przez Juliusza Cezara w roku 46 p.n.e. opracowany został przez astronoma Sosigenesa z Aleksandrii. Był on ulepszoną wersją kalendarza egipskiego, w którym wszystkie lata były jednakowe i zawierały po 365 dni. W kalendarzu juliańskim na każde cztery lata przypadały 3 lata po 365 dni i jeden rok, tzw. rok przestępny, zawierający 366 dni. Dodatkowy dzień — 29 lutego — został dołączony do roku o numerze podzielonym przez 4. Kalendarz juliański „spóźniał się” w stosunku do Słońca o trzy doby w ciągu 400 lat — tak, że w końcu XVI wieku różnica między kalendarzem i zjawiskami astronomicznymi przekraczała 10 dni.

W 1582 roku dokonano reformy kalendarza juliańskiego. Twórcą zreformowanego kalendarza, nazwanego na cześć papieża Grzegorza XIII gregoriańskim, był Krzysztof Clavius. Po 4 października 1582 roku nastąpił od razu 15 października, co — jednorazowo — zlikwidowało niezgodność kalendarza ze zjawiskami astronomicznymi. Ulepszenie na przyszłość polegało na tym, że wśród lat o numerach podzielnych przez sto przestępnymi pozostały tylko te, których numer podzielny jest przez 400. Różnica między kalendarzem gregoriańskim a rachubą lat zwrotnikowych jest tak mała, że dochodzi do jednej doby dopiero po upływie 3300 lat, co w zupełności wystarcza dla celów praktycznych.

Nie zawsze jednak wystarcza to do celów naukowych. Aby uniknąć nieporozumień związanych ze stosowaniem różnych kalendarzy oraz aby ułatwić obliczanie długich odstępów czasowych, historycy i astronomowie stosują na wielką skalę ciągłą rachubę dni — tzw. dni juliańskie. Metoda ta została opracowana w 1583 roku przez Józefa Scaligera, a po raz pierwszy zastosowana w astronomii przez Johna Herschela w XVIII wieku. Począwszy od 1 stycznia 4713 roku p.n.e. każdej dacie odpowiada kolejny numer — dzień juliański rozpoczynający się w średnie południe czasu uniwersalnego. Dzień zerowy został wybrany w tak odległej przeszłości, by przy badaniach historii dziejów ludzkości nie zaistniała potrzeba stosowania ujemnych dni. Obecnie dni juliańskie wyrażają się liczbami siedmiocyfrowymi — np. 1 stycznia 1988 roku będzie 2447162 dniem juliańskim.

Przedstawiony obok program pozwala na wzajemne przeliczanie dni juliańskich i dat. Napisany jest w języku Basic, co daje możliwość zastosowania go praktycznie na każdym mikrokomputerze. Po uruchomieniu programu użytkownik wybiera tryb pracy ($K = 0$ — przeliczanie dnia juliańskiego na datę kalendarzową, $K = 1$ — odwrotnie) i rodzaj kalendarza ($G = 0$ — juliański, $G = 1$ — gregoriański). Dane wejściowe podawane są w postaci: ROK, DZIEŃ, MIESIĄC oraz GODZINA, MINUTY, SEKUNDY czasu uniwersalnego w trybie 0 lub numer i ułamek dnia juliańskiego w trybie 1. Wyniki obliczeń drukowane są na ekranie.

Należy tu zwrócić uwagę na pewną pułapkę związaną z oznaczaniem lat p.n.e. W programie wykorzystana jest tzw. rachuba astronomiczna, która różni się od ogólnie stosowanej historycznej. Według rachuby astronomicznej 1 rok p.n.e. to rok 0, a 2 r.p.n.e. to rok -1 itd. Początek rachuby dni juliańskich wypada więc w roku -4712.

Oprócz zastosowań czysto astronomicznych wykorzystując dni juliańskie możemy łatwo określić liczbę dni między różnymi momentami unikając kłopotliwego przeliczania lat przestępnych. Jak bowiem inaczej prosto obliczyć, ile dni upłynęło od np. bitwy pod Grunwaldem czy też od Twoich, Czytelniku, urodzin?